

D8 =РУНГКУТ("DEqn";A8;B8;C8)

Решение дифференциального уравнения $y'(x) = x^2 - y(x)$ методом Рунге-Кутты

Нач. точка	Нач. значение	Аргумент	Решение	Точно
0	5	3	5,149361224	5,149361205
0	2	10	82,00000219	82
1	0	5	16,98168441	16,98168436

Рис. 15.4
Решение дифференциального уравнения методом Рунге — Кутты
с помощью пользовательской функции

Рабочий документ, в котором реализованы все вычисления, представлен на рисунке 15.4.

В ячейках столбца A вводятся значения для аргумента в начальной точке (значение параметра x_0), в ячейках столбца B вводятся значения функции в начальной точке (параметр y_0), в ячейках столбца C указываются значения аргумента x для вычисления в них значения функции $y(x)$. Приближенное решение вычисляется в ячейках столбца D, а точное решение — в ячейках столбца E. В частности, в ячейку D6 вводится формула =РУНГКУТ(“DEqn”;A6;B6;C6), а в ячейку E6 вводится формула =C6^2-2*C6+2+(B6-2+2*A6-A6^2)*EXP(A6-C6). Затем эти формулы копируются в нижние ячейки.

На заметку

Желающие могут поэкспериментировать с начальными условиями и другими параметрами вычислений, однако в целом можно заключить, что точность вычислений вполне приемлемая.

МЕТОД АДАМСА

Я мечтал об этом всю свою сознательную жизнь.

Из к/ф «Ирония судьбы, или С легким паром!»

Метод Адамса относится к *конечно-разностным методам*. В отличие от одношаговых методов (таких, как метод Эйлера или Рунге — Кутты), в которых для расчета значения функции в узловой точке используется значение этой функции в соседней точке, в конечно-разностных схемах для вычисления значения в узловой точке используют значения в нескольких соседних узловых точках. Например, в методе Адамса третьего порядка вычисление значения функции y_{n+1} в узловой точке x_{n+1} выполняется по формуле

$$y_{n+1} = y_n + \frac{h}{12}\big(23f(x_n, y_n) - 16f(x_{n-1}, y_{n-1}) + 5f(x_{n-2}, y_{n-2})\big).$$

Как и ранее, здесь речь идет о решении уравнения $y'(x) = f(x, y(x))$ с начальным условием $y(x_0) = y_0$, и использованы обозначения $h = x_{k+1} - x_k$, $x_n = x_0 + nh$, а через y_k обозначается значение (вычисляемое или вычисленное) функции $y(x)$ в точке x_k.

В методе Адамса четвертого порядка рекуррентное соотношение для определения значения искомой функции в новой узловой точке имеет вид

$$y_{n+1} = y_n + \frac{h}{24}\big(55f(x_n,\ y_n) - 59f(x_{n-1},\ y_{n-1}) +$$
$$+37f(x_{n-2},\ y_{n-2}) - 9f(x_{n-3},\ y_{n-3})\big).$$

На заметку

Очевидно, что приведенные выше соотношения не могут применяться для вычисления значения функции в нескольких начальных точках. Для вычисления функции в начальных узловых точках используют одношаговые методы: например, в точках с индексами 1 и 2 значения функции вычисляются одношаговым методом (скажем, Эйлера или Рунге — Кутты), а во всех остальных — по методу Адамса третьего порядка. Если мы хотим использовать метод Адамса четвертого порядка, то одношаговыми методами придется вычислять значения в узловых точках с индексами от 1 по 3 включительно.

Рабочий документ, в котором методом Адамса решается уравнение $y'(x) = \sin x - y(x)$ с начальным условием $y(0) = y_0$, представлен на рисунке 15.5.

Решение дифференциального уравнения *y'(x) = sinx - y(x)*
с начальным условием *y(0) = y₀* методом Адамса

Нач. значение	2
Шаг	0,01

Итерация	**Аргумент**	**Функция ур - 3**	**Адамс 3**	**Функция ур. 4**	**Адамс 4**	**Точно**
0	0	-2	2	-2	2	2
1	0,01	-1,970000167	1,98	-1,970000167	1,98	1,980149501
2	0,02	-1,940301332	1,960299998	-1,940301332	1,960299998	1,960596013
3	0,03	-1,910901485	1,940896985	-1,910901485	1,940896985	1,941336567
4	0,04	-1,881944389	1,921933724	-1,881944397	1,921933731	1,922368212
5	0,05	-1,853278051	1,90325722	-1,853277519	1,903256689	1,903688016
6	0,06	-1,824902554	1,88486656	-1,824902604	1,884866611	1,885293067
7	0,07	-1,796815353	1,8667582	-1,796815389	1,866758236	1,867180473
8	0,08	-1,769014587	1,848929281	-1,769014642	1,848929336	1,84934736
9	0,09	-1,741498397	1,831376946	-1,741498455	1,831377004	1,831790871
10	0,1	-1,714264941	1,814098357	-1,714265005	1,814098422	1,814508171
20	0,2	-1,457088051	1,655757382	-1,457088173	1,655757503	1,656128259
30	0,3	-1,226281563	1,52180177	-1,226281731	1,521801937	1,52213741
40	0,4	-1,020256694	1,409675036	-1,020256899	1,409675241	1,409978789
50	0,5	-0,837547703	1,316973242	-0,837547937	1,316973476	1,317248138
			82,15923394		82,15924022	82,1765226

Погрешность 3	0,017288658
Погрешность 4	0,017282379
П3 > П4	ИСТИНА

Рис. 15.5
Решение дифференциального уравнения методом Адамса

Мы для сравнения вычисляли решение методом Адамса как третьего, так и четвертого порядка. Кроме того, мы учли, что точное решение уравнения (с соответствующим граничным условием) имеет вид $y(x) = \frac{1}{2}\left(\sin(x) - \cos(x) + \exp(-x)(1+2y_0)\right)$, поэтому в рабочем документе также приводятся и «точные» значения в узловых точках искомой функции, вычисленные по аналитической формуле.

В столбце A, начиная с ячейки A10 (значение 0) отображаются индексы узловых точек. В столбце B вычисляются значения аргумента в узловых точках: в ячейку B10 вводится формула =A10*$B$7, и затем она копируется в нижние ячейки. Эта формула содержит абсолютную ссылку на ячейку B7, в которую вводятся значения для шага приращения аргумента.

Значения в ячейках столбца C вычисляются так: в ячейку C10 вводим формулу =SIN(B10)-D10 и копируем ее в ячейки внизу. Данная формула является реализацией выражения для функции $f(x, y) = \sin x - y$ (функция в правой части дифференциального уравнения). Формула содержит ссылку на ячейку B10 со значением аргумента x, и ссылку на ячейку D10 со значением аргумента y.

Столбец D, в котором вычисляются приближенные значения решения уравнения по методу Адамса третьего порядка, заполняется следующим образом:

- в ячейку D10 вводим формулу =B6 со ссылкой на ячейку B6, содержащую значение функции в начальной точке;
- в ячейку D11 вводится формула =D10+$B$7*(SIN(B10)-D10), которой по методу Эйлера (соотношение $y_{n+1} = y_n + hf(x_n, y_n)$) вычисляется решение для узловой точки с индексом 1. Эта формула копируется в ячейки D12 и D13;
- в ячейку D14 вводится формула =D13+$B$7*(23*C13-16*C12+5*C11)/12, представляющая собой реализацию формулы для метода Адамса третьего порядка (соотношение $y_{n+1} = y_n + \frac{h}{12}(23f(x_n, y_n) - 16f(x_{n-1}, y_{n-1}) + 5f(x_{n-2}, y_{n-2}))$). Эта формула содержит ссылки на значения функции уравнения в трех предыдущих точках. Формулу копируем в нижние ячейки.

Два соседних столбца (E и F) предназначены для вычисления приближенного решения по методу Адамса четвертого порядка. Для этого в ячейку F10 вводим формулу =SIN(B10)-F10 и копируем эту формулу в нижние ячейки.

На заметку

Формулу =SIN(B10)-F10 можно было бы не вводить в ячейку F10, а скопировать из ячейки C10. Правда, для этого в ячейку C10 нужно вводить не формулу =SIN(B10)-D10, а формулу =SIN($B10)-D10 (т. е. использовать смешанную ссылку на ячейку B10).

Вычисления по методу Адамса четвертого порядка выполняются в столбце F следующим образом:

- в ячейку F10 вводим формулу =B6, и эта формула копируется в нижние ячейки, вплоть до ячейки F13 — мы исходим из того, что нет необходимости дважды выполнять одни и те же вычисления (имеется в виду вычисление значения для функции-решения уравнения в начальных узловых точках по методу Эйлера);

- в ячейку F14 вводится формула =F13+$B$7*(55*E13-59*E12+37*E11-9*E10)/24, которой реализуется соотношение $y_{n+1} = y_n + \frac{h}{24}\big(55f(x_n, y_n) - 59f(x_{n-1}, y_{n-1}) + 37f(x_{n-2}, y_{n-2}) - 9f(x_{n-3}, y_{n-3})\big)$ для метода Адамса четвертого порядка. Эта формула копируется в нижние ячейки.

Наконец, ячейки в столбце G заполняются копированием из ячейки G10 формулы =(SIN(B10)-COS(B10)+EXP(-B10)*(1+2*$B$6))/2.

Специфика разработанного нами документа такова, что при изменении значения функции в начальной точке (ячейка B6) или шага приращения по аргументу (ячейка B7) весь документ, включая ячейки с точным значением для решения, пересчитывается автоматически.

Создадим функцию пользователя для решения уравнений методом Адамса. Используем такой же подход, как и в случае, когда мы разрабатывали функцию для решения уравнений методом Рунге — Кутты: функция $f(x, y)$ двух аргументов из правой части уравнения $y'(x) = f(x, y(x))$ описывается как метод в модуле класса проекта, и затем имя этого метода передается в качестве текстового параметра функции, в которой собственно и реализуется метод Адамса. Такой подход удобен, поскольку позволяет использовать одну и ту же функцию для решения различных уравнений.

Программный код функции АДАМС(), в которой реализованы все эти амбициозные планы, представлен в листинге 15.16.

Листинг 15.16. Функция для решения дифференциального уравнения методом Адамса

```
Function АДАМС(f, x0, y0, t, Optional method = True, Optional dx = 0.001)
 ‘ Переменная для записи количества узловых точек (кроме начальной)
 Dim n As Integer
 ‘ Индекс последней узловой точки
 n = Fix((t - x0) / dx)
 ‘ Общее количество точек не меньше пяти
 If (n < 4) Then n = 4
 ‘ Переменная для приращения аргумента
 Dim h As Double
 ‘ Значение приращения аргумента
 h = (t - x0) / n
 ‘ Индексная переменная для оператора цикла
 Dim i As Integer
 ‘ Объект для вызова метода, определяющего функцию уравнения
 Set obj = New DEquations
 ‘ Массив для записи значений узловых точек
 ReDim x(0 To n) As Double
 ‘ Начальная узловая точка
 x(0) = x0
 ‘ Массив для записи значений функции-решения в узловых точках
```

```
ReDim y(0 To n) As Double
' Начальное значение функции-решения
 y(0) = y0
' Массив для записи значений функции в правой части уравнения
ReDim Fn(0 To n) As Double
' Начальное значение функции в правой части уравнения
Fn(0) = CallByName(obj, f, VbMethod, x(0), y(0))
' Переменная для определения индекса узловой точки,
' до которой вычисления выполняются по методу Эйлера
Dim k As Integer
' Индекс узловой точки, до которой вычисления проводятся
' по методу Эйлера: 3 для метода Адамса 4-го порядка,
' и 2 для метода Адамса 3-го порядка
 If method Then k = 3 Else k = 2
' Оператор цикла для вычисления значений по методу Эйлера
 For i = 1 To k
  ' Значение функции-решения уравнения
   y(i) = y(i - 1) + h * Fn(i - 1)
  ' Значение узловой точки
   x(i) = x(i - 1) + h
  ' Значение функции в правой части уравнения
  Fn(i) = CallByName(obj, f, VbMethod, x(i), y(i))
Next i
' Определяем порядок метода Адамса
If method Then
  ' Метод Адамса 4-го порядка
   For i = k + 1 To n
    ' Значение функции-решения в узловой точке
     y(i) = y(i - 1) + h / 24 * (55 * Fn(i - 1) - 59 * _
     Fn(i - 2) + 37 * Fn(i - 3) - 9 * Fn(i - 4))
    ' Значение узловой точки
     x(i) = x(i - 1) + h
    ' Значение функции в правой части уравнения
    Fn(i) = CallByName(obj, f, VbMethod, x(i), y(i))
  Next i
  ' Метод Адамса 3-го порядка
Else
   For i = k + 1 To n
    ' Значение функции-решения в узловых точках
     y(i) = y(i - 1) + h / 12 * (23 * Fn(i - 1) - 16 * _
     Fn(i - 2) + 5 * Fn(i - 3))
    ' Значение узловых точек
     x(i) = x(i - 1) + h
    ' Значение функции в правой части уравнения
    Fn(i) = CallByName(obj, f, VbMethod, x(i), y(i))
  Next i
```

```
  End If
  ' Значение функции
  АДАМС = y(n)
End Function
```

Функция АДАМС() имеет шесть аргументов (два последних необязательные). Через f обозначена ссылка на имя метода, определяющего функцию в правой части дифференциального уравнения. Аргумент x0 обозначает начальную точку (по аргументу), в которой задается начальное условие. Значение функции-решения в начальной точке задается аргументом y0. Через t определяется значение аргумента, для которого вычисляется значение функции-решения уравнения. Это обязательные аргументы. Два необязательных (опционных) аргумента определяют порядок метода Адамса и величину для шага приращения аргумента. Аргумент method имеет по умолчанию значение True, и при таком значении аргумента вычисления производятся по методу Адамса четвертого порядка. Аргумент dx имеет значение по умолчанию 0.001 и определяет (примерный) шаг для приращения аргумента.

Количество частей, на которые разбивается интервал от начальной точки до точки, для которой вычисляется решение уравнения, записывается в переменную n и вычисляется командой n = Fix((t-x0)/dx). В данном случае мы берем целую часть от деления длины интервала t-x0 между начальной и конечной точками на величину шага приращения по аргументу dx. Кроме этого, используемый для вычисления решения метод предполагает наличие нескольких узловых точек. Мы требуем, чтобы узловых точек было не меньше пяти (т. е. количество интервалов — не меньше четырех). Поэтому в условном операторе проверяем условие n<4, и если оно выполнено, то командой n = 4 задаем минимально необходимое значение для переменной n.

Командой h = (t-x0)/n определяем шаг приращения для аргумента (расстояние между соседними узловыми точками). Помимо этого, создаем несколько массивов (в каждом из них индексация элементов выполняется от 0 до n): массив x для записи узловых точек, массив y для записи значений функции-решения уравнения, а также массив Fn для записи значений функции в правой части уравнения. В каждом из этих массивов определяется начальный (с нулевым индексом) элемент: x(0) = x0 (начальное значение для аргумента), y(0) = y0 (начальное значение функции-решения) и Fn(0) = CallByName (obj,f,VbMethod,x(0),y(0)) — данной командой вычисляется значение функции в правой части дифференциального уравнения в начальной точке (при значениях аргумента и искомой функции-решения, определяемых начальным условием). При этом объект obj, из которого вызывается метод, передаваемый по ссылке f, заранее создается командой Set obj = New DEquations.

На заметку

Таким образом, в проект необходимо добавить модуль класса с названием DEquations и описать там функцию двух переменных, которая будет определять выражение в правой части дифференциального уравнения. Имя этой функции (в текстовом формате) будет передаваться функции АДАМС() в качестве аргумента.

Значения в нескольких начальных узловых точках мы будем вычислять по методу Эйлера (формула $y_i = y_{i-1} + hf(x_i, y_i)$). Если используется метод Адамса четвертого порядка, то по методу Эйлера вычисляются значения функции-решения уравнения до узловой точки с индексом 3 включительно. Если используется метод Адамса третьего порядка, вычислять по методу Эйлера значения функции-решения следует до узловой точки с индексом 2. Значение индекса узловой точки, до которой производятся вычисления по методу Эйлера, определяем с помощью условного оператора и записываем в переменную k. Затем запускается оператор цикла с индексной переменной i, пробегающей значения от 1 до k. За каждый цикл выполняется три команды:

- командой y(i) = y(i-1)+h*Fn(i-1) вычисляется значение функции-решения в очередной узловой точке;
- значение узловой точки вычисляется командой x(i) = x(i-1)+h;
- значение функции в правой части уравнения для вычисленных значений x(i) и y(i) вычисляется по формуле Fn(i) = CallByName(obj,f,VbMethod,x(i),y(i)).

После того как по методу Эйлера вычислены значения в начальных точках, во всех последующих точках значения для функции-решения уравнения вычисляются по методу Адамса третьего или четвертого порядков — в зависимости от значения аргумента method. И в том, и в другом случае запускается оператор цикла, с командами, аналогичными рассмотренным выше. Разница только в способе вычисления значения функции-решения дифференциального уравнения в очередной узловой точке. Для метода Адамса четвертого порядка следующее значение функции вычисляется на основе данных по четырем предыдущим точкам по формуле y(i) = y(i-1)+h/24*(55*Fn(i-1)-59*Fn(i-2)+37*Fn(i-3)-9*Fn(i-4)), а для метода Адамса третьего порядка новое значение функции-решения вычисляется по формуле y(i) = y(i-1)+h/12*(23*Fn(i-1)-16*Fn(i-2)+5*Fn(i-3)), в которой используются данные вычислений по трем предыдущим узловым точкам.

Значение y(n) в последней узловой точке возвращается как результат функции АДАМС().

Помимо этой функции, нам нужно описать функцию от двух аргументов, определяющую правую часть решаемого дифференциального уравнения. Для решения уравнения $y'(x) = \sin x - y(x)$ в модуле класса DEquations описываем функцию DEFn(), задающую зависимость $f(x, y) = \sin x - y$ (листинг 15.17).

Листинг 15.17. Функция, определяющая решаемое методом Адамса уравнение

```
Function DEFn(x As Double, y As Double) As Double
  DEFn = Sin(x) - y
End Function
```

Документ, в котором с использованием разработанных и описанных выше пользовательских функций решается уравнение $y'(x) = \sin x - y(x)$ с начальным условием общего вида $y(x_0) = y_0$, представлен на рисунке 15.6.

В ячейку A7 вводится значение для аргумента x_0 (точка, в которой задается начальное условие). В ячейку B4 вводится значение искомой функции в

D7 =АДАМС("DEFn";$A$7;$B$7;C7;ЛОЖЬ)

Решение дифференциального уравнения $y'(x) = sin(x) - y(x)$ с начальным условием $y(x_0) = y_0$ методом Адамса

	A	B	C	D	E	F
6	Нач. точка	Нач. знач.	Аргумент	Адамс - 3	Адамс - 4	Точно
7	0,0	2,0	0,0	2,000000000	2,000000000	2,000000000
8			3,0	0,690023773	0,690023699	0,690023923
9			6,0	-0,613596019	-0,613596023	-0,613596012
10			9,0	0,661932898	0,661932898	0,661932898
11			12,0	-0,690198078	-0,690198078	-0,690198078

Рис. 15.6
Решение дифференциального уравнения методом Адамса
с помощью пользовательских функций

начальной точке (параметр y_0). Аргументы, для которых вычисляется решение, заносятся в ячейки C7:C11. При этом в ячейку C7 мы ввели формулу =A7, в ячейку C8 ввели формулу =C7+3, а затем скопировали ее в ячейки C9:C11 — в результате получаем несколько значений для аргумента с шагом дискретности 3. Решения (приближенные) вычисляются в ячейках диапазона D7:E11: а именно, в ячейках D7:D11 представлены значения аргументов, а в ячейках C7:C11 решение вычисляется по методу Адамса третьего порядка. Вычисления по методу Адамса четвертого порядка выполняются в ячейках E7:E11. Ячейки заполняются так: в ячейку D7 вводим формулу =АДАМС (“DEFn”;$A$7;$B$7;C7;ЛОЖЬ) и копируем ее в остальные ячейки диапазона D7:D11, а в ячейку E7 вводится формула =АДАМС(“DEFn”;$A$7;$B$7;C7) и копируется в нижние ячейки до заполнения диапазона E7:E11. Точное решение вычисляется в ячейках F7:F11 копированием формулы =(SIN(C7)-COS(C7)+EXP($A$7-C7)*(2*$B$7+COS ($A$7)-SIN($A$7)))/2 из ячейки F7 в нижние ячейки диапазона. В данном случае мы воспользовались тем, что точным решением для задачи $y'(x) = = \sin x - y(x)$ с начальным условием $y(x_0) = y_0$ является функция

$$y(x) = \frac{1}{2}\left(\sin(x) - \cos(x) + \exp(x_0 - x)(2y_0 + \cos(x_0) - \sin(x_0))\right).$$

МЕТОД МИЛНА

Это же вам не лезгинка, а твист!

Из к/ф «Кавказская пленница»

Метод Милна относится к методам типа *предиктор-корректор*, в которых сначала вычисляется приближенное значение для функции в узловой точке, а потом это значение уточняется. Мы воспользуемся следующей версией метода Милна для решения уравнения $y'(x) = f(x, y(x))$ с начальным